
objc4源码分析

理解runtime做了什么，对象到底是什么，很多问题的答案就会变得⾮常清晰，⽐如：�

• 为什么Category不能添加ivar？�

• ⽅法查找的本质是什么？

• Category的同名⽅法为什么会覆盖原类的⽅法？为什么Protocol不会？属性重名会怎么样？�

• ⼦类实现了和⽗类同名的私有⽅法会覆盖⽗类的实现吗？为什么？

• 类对象、元类对象是什么？

• 为什么可以给类对象和元类对象发送消息？

• IMP、Method、SEL的区别？�

• Method�Swizzling做了什么？�

• associatedObject到底是什么？�

这些问题在阅读过源码后就会变得很容易理解

前⾔

之前的问题中，基本上所有都和runtime有关，所以runtime是什么？运⾏时⼜是什么？runtime和运

⾏时有什么关系？

• runtime是⼀个动态连接库，被dyld加载到进程的内存空间，这个动态连接库包含了⼀系列的运⾏
时能⼒，包括类对象的定义和加载、消息发送、⽅法查找、动态⽅法解析、反射等等

• 运⾏时：⼴义上是指程序从启动到退出之间的整个⽣命周期，即程序在操作系统加载后被�CPU�执
⾏的过程

• runtime和运⾏时：runtime时运⾏时阶段所依赖的核⼼库，为Objective-C程序在运⾏时提供元数
据管理、对象模型，消息分发等所有底层能⼒

1. 类对象�

1.1 类对象是什么？�

代码块�

id object = [[NSObject alloc] init];1



• [[NSObject�alloc]�init]创建了⼀个实例，也就是⼀块NSObject的对象布局的内存地址，⽽object是
⼀个指针，指向这⼀块内存地址的头部，这也就是我们所说的持有了这个对象的引⽤

• 在objective-c中，基本上所有对象都继承⾃两个基类NSObject和NSProxy，⽽对象在运⾏时是这
样定义的

代码块�

typedef struct objc_class *Class;

/// Represents an instance of a class.
struct objc_object {
    Class _Nonnull isa  OBJC_ISA_AVAILABILITY;
};

/// A pointer to an instance of a class.
typedef struct objc_object *id;

• 所以说，所有对象都是⼀个objc_object，⽽id就是⼀个objc_object类型的指针，所以可以指向任
何对象

• 我们都说，有isa的就是对象，因为有完整的isa链就能收发消息�

super关键字�

代码块�

struct objc_super {
    /// Specifies an instance of a class.
    __unsafe_unretained _Nonnull id receiver;

    /// Specifies the particular superclass of the instance to message. 
    __unsafe_unretained _Nonnull Class super_class;

    /* super_class is the first class to search */
};

• 在objc4的定义中，super就是⼀个结构体，存储着实例和它的superClass，调⽤super的时候，会
第⼀个搜索super_class⽽不是receiver�

objc_class�

代码块�

struct objc_class : objc_object {
  objc_class(const objc_class&) = delete;
  objc_class(objc_class&&) = delete;

  void operator=(const objc_class&) = delete;
  void operator=(objc_class&&) = delete;
    // Class ISA;
    Class superclass;
    cache_t cache;             // formerly cache pointer and vtable
    class_data_bits_t bits;    // class_rw_t * plus custom rr/alloc flags

    Class getSuperclass() const {
#if __has_feature(ptrauth_calls)
#   if ISA_SIGNING_AUTH_MODE == ISA_SIGNING_AUTH
        if (superclass == Nil)
            return Nil;

#if SUPERCLASS_SIGNING_TREAT_UNSIGNED_AS_NIL
        void *stripped = ptrauth_strip((void *)superclass, ISA_SIGNING_KEY);
        if ((void *)superclass == stripped) {
            void *resigned = ptrauth_sign_unauthenticated(stripped, 
ISA_SIGNING_KEY, ptrauth_blend_discriminator(&superclass, 
ISA_SIGNING_DISCRIMINATOR_CLASS_SUPERCLASS));
            if ((void *)superclass != resigned)
                return Nil;
        }
#endif
            
        void *result = ptrauth_auth_data((void *)superclass, ISA_SIGNING_KEY, 
ptrauth_blend_discriminator(&superclass, 
ISA_SIGNING_DISCRIMINATOR_CLASS_SUPERCLASS));
        return (Class)result;

#   else
        return (Class)ptrauth_strip((void *)superclass, ISA_SIGNING_KEY);
#   endif
#else
        return superclass;
#endif
    }

    class_rw_t *data() const {
        return bits.data();
    }

• objc_class的结构体：继承⾃objc_object，所以说Class也有isa，是⼀个对象，它也可以接收+转
发消息

• 类对象isa元类对象，⽽元类对象也是⼀个Class，不同的只是元类对象存储的是类⽅法�

• 类对象和元类对象在dyld之后被runtime加载并建⽴isa、superclass关系，在进⼊main()函数之前
被全部加载和注册

• cache：存储最近查找过的methods(这⾥不准确，后⾯有解释)和protocols�

• bits：可以看到，著名的rw_t就是bits.data()�

class_rw_t

代码块�

struct class_rw_t {
    Class firstSubclass;
    Class nextSiblingClass;

private:
    const class_ro_t *ro() const {
        auto v = get_ro_or_rwe();
        if (slowpath(v.is<class_rw_ext_t *>())) {
            return v.get<class_rw_ext_t *>(&ro_or_rw_ext)->ro;
        }
        return v.get<const class_ro_t *>(&ro_or_rw_ext);
    }

    const method_array_t methods() const {
        auto alternates = methodAlternates();
        if (auto *array = alternates.array)
            return *array;
        if (auto *list = alternates.list)
            return method_array_t{list};
        if (auto *relativeList = alternates.relativeList)
            return method_array_t{relativeList};
        return method_array_t{};
    }

    const property_array_t properties() const {
        auto v = get_ro_or_rwe();
        if (v.is<class_rw_ext_t *>()) {
            return v.get<class_rw_ext_t *>(&ro_or_rw_ext)->properties;
        } else {
            auto &baseProperties = v.get<const class_ro_t *>(&ro_or_rw_ext)-
>baseProperties;
            return property_array_t{baseProperties};
        }
    }

    const protocol_array_t protocols() const {
        auto v = get_ro_or_rwe();
        if (v.is<class_rw_ext_t *>()) {

            return v.get<class_rw_ext_t *>(&ro_or_rw_ext)->protocols;
        } else {
            auto &baseProtocols = v.get<const class_ro_t *>(&ro_or_rw_ext)-
>baseProtocols;
            return protocol_array_t{baseProtocols};
        }
    }
};

• class_rw_t：也就是class_readwrite_t，存储在运⾏时动态添加的⽅法，协议，属性，可能来⾃于
Category或者运⾏时⽅法�

• 这就是为什么Category不能添加实例变量的原因，因为ivar在编译期就已经确定，被添加进了ro_t�

• setAssociatedObject并不在rw_t中，关联对象的的名字起的很好，associatedObject是通过⼀个
全局map的形式，把变量和objc_object关联�

class_ro_t�

代码块�

struct class_ro_t {
    uint32_t flags;
    uint32_t instanceStart;
    uint32_t instanceSize;
#ifdef __LP64__
    uint32_t reserved;
#endif

    union {
        const uint8_t * ivarLayout;
        Class nonMetaclass;
    };

    explicit_atomic<const char *> name;
    objc::PointerUnion<method_list_t, relative_list_list_t<method_list_t>, 
method_list_t::Ptrauth, method_list_t::Ptrauth> baseMethods;
    objc::PointerUnion<protocol_list_t, relative_list_list_t<protocol_list_t>, 
PtrauthRaw, PtrauthRaw> baseProtocols;
    const ivar_list_t * ivars;

    const uint8_t * weakIvarLayout;
    objc::PointerUnion<property_list_t, relative_list_list_t<property_list_t>, 
PtrauthRaw, PtrauthRaw> baseProperties;

    // This field exists only when RO_HAS_SWIFT_INITIALIZER is set.

    _objc_swiftMetadataInitializer __ptrauth_objc_method_list_imp 
_swiftMetadataInitializer_NEVER_USE[0];

    _objc_swiftMetadataInitializer swiftMetadataInitializer() const {
        if (flags & RO_HAS_SWIFT_INITIALIZER) {
            return _swiftMetadataInitializer_NEVER_USE[0];
        } else {
            return nil;
        }
    }
};

• 成员：baseMethods、baseProtocols、ivars、baseProperties�

• 在编译期就已经确定

1.1.1 ⽅法协议查询产⽣的问题�

先复习⼀下⽅法和协议查询，不提后续isa链和消息转发的查询流程是这样�

• objc_object通过isa找到⾃⼰的类对象，先查cache_t，再查rw_t，再查ro_t，找不到就会继续isa
链，找了就缓存进cache_t然后返回�

我们都知道，Category会覆盖原类的⽅法�

• 原因也很简单，Category中的协议，属性，⽅法都会被runtime塞进rw_t中，因为查询顺序的原
因，先查rw_t，找到就返回，所以并不是⽅法覆盖，只是有多个⽅法，找到就返回了�

这样会引出⼀个问题，protocol为什么不影响，属性会不会影响？�

• 协议的查询：conformsToProtocol会在找到后就返回，所以不会受影响，只要有就⾏，反正是⼀
样的

• 属性的查询：

代码块�

struct property_t {
    const char *name;
    const char *attributes;
};

属性的结构体只有⼀个name和attributes，attributes中S和G字符串表⽰这个属性的getter和setter具

体是哪个selector�

• 类和Category实现了同样name的属性编译器不会报错�

• runtime查询属性实现的本质其实就是查询⽅法，所以如果有同名属性，并且属性的getter或setter
⽅法名相同，也会影响⽅法查询，谁后加载就⽤谁

1.1.2 selector到底是什么？�

使⽤clang�-rewrite-objc�LabObject.m之后，可以看到⽅法调⽤的转换�

代码块�

((LabObject *(*)(id, SEL))(void *)objc_msgSend)((id)((LabObject *(*)(id, SEL))
(void *)objc_msgSend)((id)objc_getClass("LabObject"), 
sel_registerName("alloc")), sel_registerName("init"))

• sel_registerName将会返回⼀个SEL，SEL是⼀个objc_selector结构体，⽽objc_selector的本质是
⼀个字符串指针

• 相同的SEL都指向同⼀块地址，在发送消息时，唯⼀不同的就是接收对象，msgSend会根据SEL和
object找到类对象的method，调⽤imp�

• 对⽐SEL的本质是对⽐字符串指针，相同的SEL总是指向同⼀块地址，这是runtime对查询的⼀种优
化，对⽐指针的速度是极快的

这⾥要注意的是Objective-C并不关⼼参数类型，即使参数类型不同，只要名称⼀样，就会被认为是同

⼀个SEL�

代码块�

- (void)someMethod:(NSObject *)object;
- (void)someMethod:(NSProxy *)proxy;

Objective-C不会关⼼参数类型，甚⾄不会关⼼参数名，这两个⽅法会被认为是同⼀个SEL，后⾯定义

的method会覆盖前⾯的，这也就是Objective-C不⽀持重载的原因�

2. Method�Swizzling的本质是什么�

代码块�

SEL originalSelector = NSSelectorFromString(selectors[i]);
SEL swizzledSelector = NSSelectorFromString(swizzleds[i]);

Method originalMethod = class_getInstanceMethod(class, originalSelector);
Method swizzledMethod = class_getInstanceMethod(class, swizzledSelector);

if (!originalMethod || !swizzledMethod) continue;

BOOL didAddMethod =
class_addMethod(class,
                originalSelector,
                method_getImplementation(swizzledMethod),
                method_getTypeEncoding(swizzledMethod));

if (didAddMethod) {
    class_replaceMethod(class,
                        swizzledSelector,
                        method_getImplementation(originalMethod),
                        method_getTypeEncoding(originalMethod));
} else {
    method_exchangeImplementations(originalMethod, swizzledMethod);
}

具体逻辑简要分析如下

1. 通过SEL找到Method�

2. 尝试将swizzlingMethod的实现添加到originalSelector上�

3. 如果添加上了，替换sizzledMethod的实现�

4. 如果没有，交换两个⽅法的实现

所以最关键的⽅法就是class_addMethod、class_replaceMethod、

method_exchangeImplementations�

代码块�

BOOL
class_addMethod(Class cls, SEL name, IMP imp, const char *types)
{
    if (!cls) return NO;

    mutex_locker_t lock(runtimeLock);
    return ! addMethod(cls, name, imp, types ?: "", NO);
}

IMP
class_replaceMethod(Class cls, SEL name, IMP imp, const char *types)
{
    if (!cls) return nil;

    mutex_locker_t lock(runtimeLock);
    return addMethod(cls, name, imp, types ?: "", YES);
}

可以看到addMethod和replaceMethod调⽤了同⼀个⽅法，只是最后传⼊的参数不同�

然⽽在objc-runtime-new.mm⽂件中，似乎并没有addMethod这个⽅法的准确实现，只有定义�

static IMP addMethod(Class cls, SEL name, IMP imp, const char *types, bool 
replace);

method_exchangeImplementations�

代码块�

void method_exchangeImplementations(Method m1Signed, Method m2Signed)
{
    if (!m1Signed  ||  !m2Signed) return;

    method_t *m1 = _method_auth(m1Signed);
    method_t *m2 = _method_auth(m2Signed);

    mutex_locker_t lock(runtimeLock);

    IMP imp1 = m1->imp(false);
    IMP imp2 = m2->imp(false);
    SEL sel1 = m1->name();
    SEL sel2 = m2->name();

    m1->setImp(imp2);
    m2->setImp(imp1);

    flushCaches(nil, __func__, [sel1, sel2, imp1, imp2](Class c){
        return c->cache.shouldFlush(sel1, imp1) || c->cache.shouldFlush(sel2, 
imp2);
    });

    adjustCustomFlagsForMethodChange(nil, m1);
    adjustCustomFlagsForMethodChange(nil, m2);
}

• 可以看到⽅法交换的本质就是IMP交换�

• 最后会更新缓存，因为缓存⾥的SEL可能还是指向原来的imp�

• 所以由此可知，objc_class中的cache_t实际上缓存的是SEL到IMP的映射，⽽不是Method�

• 最后会更新⽅法的flag，flag会表⽰该method的实现是不是被runtime替换过�

3. AssociatedObject的本质是什么？�

关联对象其实并不复杂

代码块�

id
objc_getAssociatedObject(id object, const void *key)
{
    return _object_get_associative_reference(object, key);
}

void
objc_setAssociatedObject(id object, const void *key, id value, 
objc_AssociationPolicy policy)
{
    _object_set_associative_reference(object, key, value, policy);
}

• 所以重点就是这两个⽅法_object_get_associative_reference和
_object_set_associative_reference�

_object_get_associative_reference�

代码块�

id
_object_get_associative_reference(id object, const void *key)
{
    ObjcAssociation association{};

    {
        AssociationsManager manager;
        AssociationsHashMap &associations(manager.get());
        AssociationsHashMap::iterator i = associations.find((objc_object 
*)object);
        if (i != associations.end()) {
            ObjectAssociationMap &refs = i->second;
            ObjectAssociationMap::iterator j = refs.find(key);
            if (j != refs.end()) {
                association = j->second;
                association.retainReturnedValue();
            }
        }
    }

    return association.autoreleaseReturnedValue();
}

先了解⼀下基本的数据结构的定义

class ObjcAssociation {
    uintptr_t _policy;
    id _value;
}

typedef DenseMap<const void *, ObjcAssociation> ObjectAssociationMap;

typedef DenseMap<DisguisedPtr<objc_object>, ObjectAssociationMap> 
AssociationsHashMap;

• ObjcAssociation：代表⼀条关联对象的记录，包含policy和value�

• AssociationsHashMap：object到ObjectAssociationMap之间的映射，是⼀个全局map�

• ObjectAssociationMap：key到ObjectAssociation之间的映射�

接下来分析源码

• 声明⼀个空的ObjcAssociation结构体�

• AssociationsManager �是⼀个�RAII（⾃动资源管理）对象，进⼊作⽤域就加锁，析构时⾃动
解锁，确保线程安全。

• 从manager中获取全局的关联对象哈希表，这⾥存储在所有关联对象的映射�

代码块�

AssociationsHashMap::iterator i = associations.find((objc_object *)object);

• 查找关联对象在不在map⾥，i�!=�end代表找到了，后⾯的i->second相关的操作都是c++⾥的取
value�

代码块�

class ObjcAssociation {
    uintptr_t _policy;
    id _value;
public:
    inline void retainReturnedValue() {
        if (_value && (_policy & OBJC_ASSOCIATION_GETTER_RETAIN)) {
            objc_retain(_value);
        }
    }

    inline id autoreleaseReturnedValue() {
        if (slowpath(_value && (_policy & 
OBJC_ASSOCIATION_GETTER_AUTORELEASE))) {
            return objc_autorelease(_value);

        }
        return _value;
    }
};

typedef OBJC_ENUM(uintptr_t, objc_AssociationPolicy) {
    OBJC_ASSOCIATION_ASSIGN = 0,           /**< Specifies an unsafe unretained 
reference to the associated object. */
    OBJC_ASSOCIATION_RETAIN_NONATOMIC = 1, /**< Specifies a strong reference 
to the associated object. 
                                            *   The association is not made 
atomically. */
    OBJC_ASSOCIATION_COPY_NONATOMIC = 3,   /**< Specifies that the associated 
object is copied. 
                                            *   The association is not made 
atomically. */
    OBJC_ASSOCIATION_RETAIN = 01401,       /**< Specifies a strong reference 
to the associated object.
                                            *   The association is made 
atomically. */
    OBJC_ASSOCIATION_COPY = 01403          /**< Specifies that the associated 
object is copied.
                                            *   The association is made 
atomically. */
};

• 内存管理语义，如果set的时候指定了retain，找到后先增加引⽤计数，防⽌返回前对象被释放�

• slowpath(...) ：这是�clang/gcc�的�branch�prediction�宏，提⽰编译器这个条件不常⾛（性
能优化，不影响逻辑）

• atomic修饰的属性会默认返回autorelease对象，所以在使⽤set管理对象的时候，policy就已经决
定了get关联对象需不需要返回autorelease对象�

总结

• getAssociatedObject本质是就是在⼀个全局的map中根据object找到这个object关联的所有对
象，然后根据key返回这个关联对象的值，如果是atomic修饰符修饰的会返回autorelease对象�

_object_set_associative_reference�

代码块�

void
_object_set_associative_reference(id object, const void *key, id value, 
uintptr_t policy)
{
    // This code used to work when nil was passed for object and key. Some code

    // probably relies on that to not crash. Check and handle it explicitly.
    // rdar://problem/44094390
    if (!object && !value) return;

    Class objectClass = object->getIsa();
    objectClass->realizeIfNeeded();

    if (objectClass->forbidsAssociatedObjects())
        _objc_fatal("objc_setAssociatedObject called on instance (%p) of class 
%s which does not allow associated objects", object, 
object_getClassName(object));

    DisguisedPtr<objc_object> disguised{(objc_object *)object};
    ObjcAssociation association{policy, value};

    // retain the new value (if any) outside the lock.
    association.acquireValue();

    bool isFirstAssociation = false;
    {
        AssociationsManager manager;
        AssociationsHashMap &associations(manager.get());

        if (value) {
            auto refs_result = associations.try_emplace(disguised, 
ObjectAssociationMap{});
            if (refs_result.second) {
                /* it's the first association we make */
                isFirstAssociation = true;
            }

            /* establish or replace the association */
            auto &refs = refs_result.first->second;
            auto result = refs.try_emplace(key, std::move(association));
            if (!result.second) {
                association.swap(result.first->second);
            }
        } else {
            auto refs_it = associations.find(disguised);
            if (refs_it != associations.end()) {
                auto &refs = refs_it->second;
                auto it = refs.find(key);
                if (it != refs.end()) {
                    association.swap(it->second);
                    refs.erase(it);
                    if (refs.size() == 0) {
                        associations.erase(refs_it);

                    }
                }
            }
        }
    }

    // Call setHasAssociatedObjects outside the lock, since this
    // will call the object's _noteAssociatedObjects method if it
    // has one, and this may trigger +initialize which might do
    // arbitrary stuff, including setting more associated objects.
    if (isFirstAssociation)
        object->setHasAssociatedObjects();

    // release the old value (outside of the lock).
    association.releaseHeldValue();
}

• 确保类已经realize：获取对象的实际类，并确保�class�已经被�runtime�realize（⽐如�lazy-load�
类）。不�realize�可能导致�class�的元数据（如�associated�flag、property�table）未初始化�

• 如果某些对象禁⽌关联对象，报错

• 根据policy和value初始化ObjcAssociation�

• association.acquireValue()根据策略retain或者copy新值�

• 区分value有没有值的情况处理设置或者删除�

• 有值：直接写⼊，先在全局表查找/插⼊�object�的�map，如果是新�object（第⼆返回�true），标
记⾸个关联。然后再插⼊/查找这个�object�的�key-value，如果已有，直接�swap，回收旧值。�

◦ try_emplace �是�C++17�map�的优化⽤法：只有�key�不存在才会�emplace，省⼀次构造和查
找。

• 没有值：把旧的ObjcAssociation取出来，swap到临时变量，后续回收，如果该object的关联对象
全部回收了，删除object相关的关联表�

• 如果是第⼀个关联对象，设置runtime状态位，通知对象�

• 释放旧的关联对象

49
50
51
52
53
54
55
56
57
58
59
60
61
62
63
64
65


